Здравствуйте, есть справочник Клиенты в нем поле Оклад, регистр СоглашенияСКлиентамиСрезПоследних (указывает на какой способ оплаты выбран, но нам интересует только "постоянная оплата")

С помощью СКД формирую запрос

```1c
    ВЫБРАТЬ
    Клиенты.Ссылка КАК Ссылка,
    СоглашенияСКлиентамиСрезПоследних.Оклад КАК Оклад
    ИЗ
    Справочник.Клиенты КАК Клиенты
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ РегистрСведений.СоглашенияСКлиентами.СрезПоследних КАК СоглашенияСКлиентамиСрезПоследних
    ПО (Клиенты.Ссылка = СоглашенияСКлиентамиСрезПоследних.Клиент)
    ГДЕ
    СоглашенияСКлиентамиСрезПоследних.СпособОплаты = &СпособОплаты
```

Он показывает контрагента и его оклад.
Иванов 1000
Петров 5000
Сидоров 6000

Есть документ поступление денежных средств + , которое делает движение в регистр накопления ВзаиморасчетыСКлиентами.

Как мне изменить запрос чтобы добавить еще одни столбик в котором будет указано что оплатил или нет за текущий месяц?

Иванов 1000 Оплатил
Петров 5000 нет
Сидоров 6000 Оплатил
---
```1c
    ВЫБРАТЬ
    Клиенты.Ссылка КАК Ссылка,
    СоглашенияСКлиентамиСрезПоследних.Оклад КАК Оклад
    ПОМЕСТИТЬ ВТСотрудники
    ИЗ
    Справочник.Клиенты КАК Клиенты
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ РегистрСведений.СоглашенияСКлиентами.СрезПоследних КАК СоглашенияСКлиентамиСрезПоследних
    ПО (Клиенты.Ссылка = СоглашенияСКлиентамиСрезПоследних.Клиент)
    ГДЕ
    СоглашенияСКлиентамиСрезПоследних.СпособОплаты = &СпособОплаты
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    ВТСотрудники.Ссылка,
    ВТСотрудники.Оклад,
    ВЫБОР
    КОГДА ЕСТЬNULL(ВзаиморасчетыСКлиентами.Клиент, 0) = 0
    ТОГДА "НЕ ПЛАТИЛ"
    ИНАЧЕ "ПЛАТИЛ"
    КОНЕЦ КАК Оплата
    ИЗ
    ВТСотрудники КАК ВТСотрудники
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрНакопления.ВзаиморасчетыСКлиентами.Обороты(&НачалоПериода, &КонецПериода, Месяц, ) КАК ВзаиморасчетыСКлиентамиОбороты
    ПО ВТСотрудники.Ссылка = ВзаиморасчетыСКлиентамиОбороты.Клиент
```
---
Какие измерения и ресурсы у РН ДеньгиПриходРасход?
---
Если я правильно понял, тогда свой запрос помещай во временную таблицу, которую соединяй левым соединением по сотруднику с регистром накопления "ДеньгиПриходРасход", у которого определи условия, типа "ТекущийМесяц" и добавь поле с условием, типа:

```1c
    Выбор
    Когда ДеньгиПриходРасход.Регистратор = Null Тогда "Не оплатил"
    Иначе "Оплатил"
    Конец
```

Соответственно, если у тебя есть еще какие-то требования к тому, оплатил он или нет, равна/превысила ли сумма поступления средств определенную отметку или нет, то их, естесственно, тоже лучше включить в условие.
---
```1c
    ВЫБРАТЬ Ссылка,Оклад,
    //Проверяем оплату
    ВЫБОР КОГДА Оплата > 0 Тогда "Оплатил" ИНАЧЕ "Не оплатил" КОНЕЦ Оплата
    ИЗ
    Справочник.Клиенты
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ (ВЫБРАТЬ Оклад, Клиент ИЗ РегистрСведений.СоглашенияСКлиентами.СрезПоследних(&КонецПериода,СпособОплаты = &СпособОплаты) ) Оклады
    ПО (Ссылка = Клиент)
    ЛЕВОЕ СОЕДИНЕНИЕ (ВЫБРАТЬ ПолеСоединения, СУММА(ПолеОборот) Оплата ИЗ РегистрНакопления.ДеньгиПриходРасход.ОстаткиИОбороты(&НачалоПериода, &КонецПериода, , , ) СГРУППИРОВАТЬ ПО ПолеСоединения) Деньги
    ПО Ссылка = ПолеСоединения
    ГДЕ СпособОплаты = &СпособОплаты
```

Запрос должен быть аналогичным, но может варьироваться в зависимости от полей регистра и вида регистра
---
Немного напутал с названием регистра накопления, приложил его структуру
---
```1c
    ВЫБРАТЬ
    Клиенты.Ссылка КАК Ссылка,
    СоглашенияСКлиентамиСрезПоследних.Оклад КАК Оклад
    ПОМЕСТИТЬ ВТСотрудники
    ИЗ
    Справочник.Клиенты КАК Клиенты
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ РегистрСведений.СоглашенияСКлиентами.СрезПоследних КАК СоглашенияСКлиентамиСрезПоследних
    ПО (Клиенты.Ссылка = СоглашенияСКлиентамиСрезПоследних.Клиент)
    ГДЕ
    СоглашенияСКлиентамиСрезПоследних.СпособОплаты = &СпособОплаты
    ;

    ////////////////////////////////////////////////////////////­////////////////////
    ВЫБРАТЬ
    ВТСотрудники.Ссылка,
    ВТСотрудники.Оклад,
    ВЫБОР
    КОГДА ЕСТЬNULL(ВзаиморасчетыСКлиентами.Клиент, 0) = 0
    ТОГДА "НЕ ПЛАТИЛ"
    ИНАЧЕ "ПЛАТИЛ"
    КОНЕЦ КАК Оплата
    ИЗ
    ВТСотрудники КАК ВТСотрудники
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрНакопления.ВзаиморасчетыСКлиентами.Обороты(&НачалоПериода, &КонецПериода, Месяц, ) КАК ВзаиморасчетыСКлиентамиОбороты
    ПО ВТСотрудники.Ссылка = ВзаиморасчетыСКлиентамиОбороты.Клиент
```
---
Может все же РегистрНакопления.ВзаиморасчетыСКлиентами.Остатки?
---
Речь вроде о движениях
Можно еще так, с указанием суммы оплаты:

```1c
    ВЫБРАТЬ
    Клиенты.Ссылка КАК Ссылка,
    СоглашенияСКлиентамиСрезПоследних.Оклад КАК Оклад
    ПОМЕСТИТЬ ВТСотрудники
    ИЗ
    Справочник.Клиенты КАК Клиенты
    ВНУТРЕННЕЕ СОЕДИНЕНИЕ РегистрСведений.СоглашенияСКлиентами.СрезПоследних КАК СоглашенияСКлиентамиСрезПоследних
    ПО (Клиенты.Ссылка = СоглашенияСКлиентамиСрезПоследних.Клиент)
    ГДЕ
    СоглашенияСКлиентамиСрезПоследних.СпособОплаты = &СпособОплаты
    ;

    ////////////////////////////////////////////////////////////­­////////////////////
    ВЫБРАТЬ
    ВТСотрудники.Ссылка,
    ВТСотрудники.Оклад,
    ВЫБОР
    КОГДА ЕСТЬNULL(ВзаиморасчетыСКлиентами.Сумма, 0) < ВТСотрудники.Оклад
    ТОГДА "НЕ ОПЛАТИЛ"
    ИНАЧЕ "ПЛАТИЛ"
    КОНЕЦ КАК Оплата,
    ЕСТЬNULL(ВзаиморасчетыСКлиентами.Сумма, 0) КАК СуммаОплаты
    ИЗ
    ВТСотрудники КАК ВТСотрудники
    ЛЕВОЕ СОЕДИНЕНИЕ РегистрНакопления.ВзаиморасчетыСКлиентами.Обороты(НАЧАЛОПЕРИОДА(&Период, МЕСЯЦ), КОНЕЦПЕРИОДА(&Период, МЕСЯЦ) Месяц, ) КАК ВзаиморасчетыСКлиентамиОбороты
    ПО ВТСотрудники.Ссылка = ВзаиморасчетыСКлиентамиОбороты.Клиент
```
---
Запятую потерял:

РегистрНакопления.ВзаиморасчетыСКлиентами.Обороты(НАЧАЛОПЕРИОДА(&Период, МЕСЯЦ), КОНЕЦПЕРИОДА(&Период, МЕСЯЦ), Месяц, )
---
Спасибо. Решение найдено, скинул награду!